* Replication package for: Catch, Restrict, and Release: The Real Story of Bank Bailouts by Allen N. Berger, Simona Nistor, Steven Ongena and Sergey Tsyplakov, Review of Corporate Finance Studies. 
* Some of the bank-level data employed in the paper are obtained from private databases and cannot be shared. The replication package contains the do-file that was used to obtain the tables in the paper corresponding to the empirical analysis. 



* This do file replicates the statistics (Tables 1-5) and the empirical regressions (Tables 6-10). The dataset allows replication of all columns in Tables 1-10. For confidentiality reasons, the original bank-level information, which is based on data from Bankscope and Orbis, has been changed. Some of the analyses make use of Stata user-contributed commands, which may have to be installed (if not installed, a Stata error message will show up). 



****************************************************************************************************************************************************************
**************************************************************************** Setup *****************************************************************************
****************************************************************************************************************************************************************

* Set the file directory where all data files should be stored (replace "Folder name" with the path of your directory, e.g., "C:\RCFS\Data")
global CRR_folder `"CRR_directory"'

cd "$RCFS_folder"


****************************************************************************************************************************************************************
*********************************************************************** Data preparation ***********************************************************************
****************************************************************************************************************************************************************

use CRR_data, clear


****************************************************************************************************************************************************************
* Compute the Harshness indices
****************************************************************************************************************************************************************

* Compute Harshness Index (Unequal weights) using PCA (Principal Component Analysis) based on the following five indicators: Dummy dividend bans, Dummy regulatory fees, Dummy board intrusions, Dummy executive pay limits, and Dummy enforcement actions
pca d_risk_based_fee_cio dsupboardint_cio dmgpayres_cio dkpayres_cio repayment_cond_restruct_cio
predict ipca_cio5
label variable ipca_cio5 "Harshness index CI  (PCA)"

pca d_risk_based_fee_dgo dsupboardint_dgo dmgpayres_dgo dkpayres_dgo repayment_cond_restruct_dgo
predict ipca_dgo5
label variable ipca_dgo5 "Harshness index DG  (PCA)"




* Compute Harshness Index (Equal weights) as the sum of the following five indicators: Dummy dividend bans, Dummy regulatory fees, Dummy board intrusions, Dummy executive pay limits, and Dummy enforcement actions
gen iew_cio5 = d_risk_based_fee_cio  + dsupboardint_cio + dmgpayres_cio + dkpayres_cio + repayment_cond_restruct_cio
label variable iew_cio5 "Harshness index CI  (Equal weights)"

gen iew_dgo5 = d_risk_based_fee_dgo  + dsupboardint_dgo + dmgpayres_dgo + dkpayres_dgo + repayment_cond_restruct_dgo
label variable iew_dgo5 "Harshness index DG  (Equal weights)"




* Compute dummy variables for sub-groups (banks with harsh restrictions versus soft restrictions)
egen ipca_cio5_med=median( ipca_cio5)
gen Dipca_cio5=0 if ipca_cio5!=.
replace Dipca_cio5=1 if ipca_cio5>= ipca_cio5_med & ipca_cio5!=.
label variable Dipca_cio5 "Dummy harsh restrictions CI (1 above median, 0 otherwise)"

egen ipca_dgo5_med=median( ipca_dgo5)
gen Dipca_dgo5=0 if ipca_dgo5!=.
replace Dipca_dgo5=1 if ipca_dgo5>= ipca_dgo5_med & ipca_dgo5!=.
label variable Dipca_dgo5 "Dummy harsh restrictions DG (1 above median, 0 otherwise)"





****************************************************************************************************************************************************************
* Data transformation
****************************************************************************************************************************************************************

* Compute natural logarithm of Total assets
gen ln_Btas = log(Btas)
label variable ln_Btas "Size (Natural logarithm of TA)"




* Compute standard deviation of ROAA and ROAE (four quartetrs rolling window of ROAA and ROAE)
tsegen Broaa_std4= rowsd(L(1/4).Broaa)
label variable Broaa_std4 "Standard deviation of ROAA 4 quarters rolling (%)"

tsegen Broae_std4= rowsd(L(1/4).Broae)
label variable Broae_std4 "Standard deviation of ROAE 4 quarters rolling (%)"




* Winsorize variables

winsor Beqta, gen(wBeqta) p(0.01)
label variable wBeqta "Equity to Total assets ratio (%)"

winsor Blcd, gen(wBlcd) p(0.01)
label variable wBlcd "Loans to Deposits (%)"

winsor ln_Btas, gen(wln_Btas) p(0.01)
label variable wln_Btas "Size (Natural logarithm of TA) (%)"

winsor Bcredit_risk, gen(wBcredit_risk) p(0.01)
label variable wBcredit_risk "Credit risk ratio (%)"

winsor Broaa, gen(wBroaa) p(0.01)
label variable wBroaa "ROA (%)"

winsor Broaa_std4, gen(wBroaa_std4) p(0.01)
label variable wBroaa_std4 "Standard deviation of ROAA (%)"

winsor Pbtas_ctas, gen(wPbtas_ctas) p(0.01)
label variable wPbtas_ctas "Percent of the national TA held by the bank (%)"

winsor Bdepta, gen(wBdepta) p(0.01)
label variable wBdepta "Deposit liabilities to Total assets (%)"

winsor Bdepliab, gen(wBdepliab) p(0.01)
label variable wBdepliab "Deposit liabilities to Total liabilities (%)"

winsor Broae_std4, gen(wBroae_std4) p(0.01)
label variable wBroae_std4 "Standard deviation of ROAE (%)"




* Generate dummy variables for year & countries
tab year, gen(Dyear)
tab countries, gen(Dcountries)




* Generate lagged variables
gen l_Beqta=l.Beqta
label variable l_Beqta "Equity to Total assets ratio lagged (%)"

gen l_wBlcd=l.wBlcd
label variable l_wBlcd "Loans to Deposits lagged (%)"

gen l_wln_Btas=l.wln_Btas
label variable l_wln_Btas "Size (Natural logarithm of TA) lagged (%)"

gen l_wBcredit_risk=l.wBcredit_risk
label variable l_wBcredit_risk "Credit risk ratio lagged (%)"

gen l_wBroaa=l.wBroaa
label variable l_wBroaa "ROA lagged (%)"

gen l_wBroaa_std4=l.wBroaa_std4
label variable l_wBroaa_std4 "Standard deviation of ROA lagged (%)"




* Sample
*Restricted sample: banks affected by either Capital injections or Liquidity bailouts
. gen Intervened_2triggers_or = (BCI1D==1) | ( BDG1D ==1)
. label variable Intervened_2triggers_or "Dummy 1 if any form of intervention (CI or DG) has been used, 0 otherwise"

*Restricted sample: banks affected by both Capital injections and Liquidity bailouts
. gen Intervened_2triggers_both = (BCI1D==1) & ( BDG1D ==1)
. label variable Intervened_2triggers_both "Dummy 1 if both forms of intervention (CI and DG) has been used, 0 otherwise"









****************************************************************************************************************************************************************
************************************************************************** Statistics **************************************************************************
****************************************************************************************************************************************************************

use CRR_data, clear



****************************************************************************************************************************************************************
* Table 1. Summary of bailout methods implemented by European banks during 2008-2014
****************************************************************************************************************************************************************

tabstat BCI  if BCId==1, by( bankname ) stats (count)
tabstat BCI  if BCId==1, by( bankname ) stats (sum)
tabstat BCITAp  if BCId==1, by( bankname ) stats (mean)


tabstat BDG  if BDGd==1, by( bankname ) stats (count)
tabstat BDG  if BDGd==1, by( bankname ) stats (sum)   
tabstat BDGTAp  if BDGd==1, by( bankname ) stats (mean)





****************************************************************************************************************************************************************
* Table 2. Comparative statistics for catch and release capital triggers across bailout methods
****************************************************************************************************************************************************************


* Catch trigger
* Capital bailouts
. tabstat l_Beqta   if BCId==1, stats(N mean median min max) col(stats)
* Debt guarantee bailouts
. tabstat l_Beqta   if BDGd==1, stats(N mean median min max) col(stats)


* Release trigger
* Capital bailouts
. tabstat l_Beqta   if BCI_release_lr==1, stats(N mean median min max) col(stats)
* Debt guarantee bailouts
. tabstat l_Beqta   if BDG_release_lr==1, stats(N mean median min max) col(stats)


***t-test
* Catch trigger
. gen t1_l_Beqta=.
. gen t2_l_Beqta=.
. replace t1_l_Beqta=l_Beqta if BCId==1
. replace t2_l_Beqta=l_Beqta if BDGd==1
. ttest t1_l_Beqta=t2_l_Beqta, unpaired


***t-test
* Release trigger
. gen t1r_l_Beqta=.
. gen t2r_l_Beqta=.
. replace t1r_l_Beqta=l_Beqta if BCI_release_lr==1
. replace t2r_l_Beqta=l_Beqta if BDG_release_lr==1
. ttest t1r_l_Beqta=t2r_l_Beqta, unpaired





****************************************************************************************************************************************************************
* Table 3. Restriction measures: Descriptive statistics
****************************************************************************************************************************************************************


* Catch trigger
* Capital bailouts
tabstat ipca_cio5 iew_cio5 if  BCId==1, stats (N mean sd median min max) col(stats)
*Debt guarantee bailouts
tabstat ipca_dgo5 iew_dgo5 if  BDGd==1, stats (N mean sd median min max) col(stats)


* Release trigger
* Capital bailouts
tabstat ipca_cio5 iew_cio5 if  BCI_release_ext==1, stats (N mean sd median min max) col(stats)
* Debt guarantee bailouts
tabstat ipca_dgo5 iew_dgo5 if  BDG_release_ext==1, stats (N mean sd median min max) col(stats)


* t-tests
gen ipca_cio5_t=.
replace ipca_cio5_t= ipca_cio5 if  BCId==1
gen ipca_dgo5_t=.
replace ipca_dgo5_t= ipca_dgo5 if  BDGd==1
ttest ipca_cio5_t== ipca_dgo5_t, unpaired

gen iew_cio5_t=.
replace iew_cio5_t= iew_cio5 if  BCId==1
gen iew_dgo5_t=.
replace iew_dgo5_t= iew_dgo5 if  BDGd==1
ttest iew_cio5_t == iew_dgo5_t , unpaired


gen ipca_cio5_r=.
replace ipca_cio5_r= ipca_cio5 if  BCI_release_ext==1
gen ipca_dgo5_r=.
replace ipca_dgo5_r= ipca_dgo5 if  BDG_release_ext==1
ttest ipca_cio5_r== ipca_dgo5_r, unpaired

gen iew_cio5_r=.
replace iew_cio5_r= iew_cio5 if  BCI_release_ext==1
gen iew_dgo5_r=.
replace iew_dgo5_r= iew_dgo5 if  BDG_release_ext==1
ttest iew_cio5_r == iew_dgo5_r , unpaired






****************************************************************************************************************************************************************
* Table 4. Catch triggers, release triggers and restrictions
****************************************************************************************************************************************************************


* Capital bailouts
* Catch trigger
. tabstat l_Beqta  if BCId==1, stats(N mean median min max) col(stats) by(Dipca_cio5)
. ttest l_Beqta  if BCId==1, by(Dipca_cio5)

* Release trigger
. tabstat l_Beqta  if BCI_release_n==1, stats(N mean median min max) col(stats) by(Dipca_cio5) 
. ttest l_Beqta  if BCI_release_n==1, by(Dipca_cio5)


* Debt guarantee bailouts
* Catch trigger
. tabstat l_Beqta  if BDGd==1, stats(N mean median min max) col(stats) by(Dipca_dgo5)
. ttest l_Beqta  if BDGd==1, by(Dipca_dgo5)


* Release trigger
. tabstat l_Beqta  if BDG_release_n==1, stats(N mean median min max) col(stats) by(Dipca_dgo5)
. ttest l_Beqta  if BDG_release_n==1, by(Dipca_dgo5)





****************************************************************************************************************************************************************
* Table 5. Bank characteristics across catch and release events: Descriptive statistics
****************************************************************************************************************************************************************


local M0 BCId BDGd
local M1 DBCI_release_ext DBDG_release_ext
local M2 Beqta Blcd ln_Btas Bcredit_risk Broaa Broaa_std4 
local M3 Pbtas_ctas Bdepta Bdepliab Broae_std4


* Catch events
tabstat `M0'  if year>=2008 , stats (N mean median min max) col(stats)
tabstat `M2' `M3'  if year>=2008 & BCId==1, stats (N mean sd median min max) col(stats)
tabstat `M2' `M3'  if year>=2008 & BDGd==1, stats (N mean sd median min max) col(stats)
tabstat `M2' `M3'  if year>=2008 & BCId!=1 & BDGd!=1, stats (N mean sd median min max) col(stats)


* Release events
* Y
tabstat DBCI_release_ext DBDG_release_ext  if year>=2008 & D_FILRtime_n2==1, stats (N mean median min max) col(stats)

* Duration
tabstat duration_CI if BCI_release_ext==1, stats (N mean sd median min max) col(stats)
tabstat duration_DG if BDG_release_ext==1, stats (N mean sd median min max) col(stats)

* Controls
tabstat `M2' `M3'  if BCI_release_ext==1, stats (N mean median min max) col(stats)
tabstat `M2' `M3'  if BDG_release_ext==1, stats (N mean median min max) col(stats)
tabstat `M2' `M3'  if year>=2008 & D_FILRtime_n2==1 & BCI_release_ext!=1 & BDG_release_ext!=1, stats (N mean median min max) col(stats)
















****************************************************************************************************************************************************************
*********************************************************************** Empirical models ***********************************************************************
****************************************************************************************************************************************************************

use CRR_data, clear


****************************************************************************************************************************************************************
* Table 6
****************************************************************************************************************************************************************

* H1: Equity to Total assets and probability to "catch" a banks through a CI or DG bailout

local M1 l.wBeqta l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4
local M4 l.wBeqta
local M5 l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4 

* Logit coefficients
logit BCId `M1' Dyear* Dcountries* if year>=2008,   cluster(banks)
est sto m
outreg2 using H1.Results.csv, replace bdec(3) sdec(3)  label addtext("Year fixed effects", YES, "Country fixed effects", YES) addstat("Pseudo R-squared", e(r2_p), "Log-likelihood", e(ll)) keep(`M4' `M5') groupvar(Catch_ratios `M4' Controls `M5') cti(Logit coeff.)

logit BDGd `M1' Dyear* Dcountries* if year>=2008,   cluster(banks)
est sto m
outreg2

* test of coeff Logit
logit BCId `M1' Dyear* Dcountries* if year>=2008
est sto m1
logit BDGd `M1' Dyear* Dcountries* if year>=2008
est sto m2
suest m1 m2
test [m1_BCId]L.wBeqta  = [m2_BDGd]L.wBeqta


* MLM Coefficients
local M1 l.wBeqta l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4 
local M5 l.wBeqta l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4  l.wBdepta  
mlogit Bailout_2trig_all `M1' Dyear* Dcountries* if year>=2008 & Bailout_2trig_all!=3,   cluster(banks) iterate(10)
est sto m
outreg2 using H1.Results.csv, append bdec(3) sdec(3)  label addtext("Year fixed effects", YES, "Country fixed effects", YES) addstat("Pseudo R-squared", e(r2_p), "Log-likelihood", e(ll)) keep(`M4' `M5') groupvar(Catch_ratios `M4' Controls `M5') cti(MLN coeff.)
mlogit
* test of coeff MLN
mlogit Bailout_2trig_all `M1' Dyear* Dcountries* if year>=2008 & Bailout_2trig_all!=3
test [Liquidity]L.wBeqta  = [Capital]L.wBeqta


* Ordered Logit Coefficients
local M1 l.wBeqta l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4 
local M5 l.wBeqta l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4  l.wBdepta  
ologit Bailout_2trig_all_o `M1' Dyear* Dcountries* if year>=2008,   cluster(banks) 
est sto m
outreg2 using H1.Results.csv, append bdec(3) sdec(3)  label addtext("Year fixed effects", YES, "Country fixed effects", YES) addstat("Pseudo R-squared", e(r2_p), "Log-likelihood", e(ll)) keep(`M4' `M5') groupvar(Catch_ratios `M4' Controls `M5') cti(Ordered Logit coeff.)

  



****************************************************************************************************************************************************************
* Table 7
****************************************************************************************************************************************************************

* H2: Equity to Total assets and probability to "release" bailed-out banks

local M1 l.wBeqta l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4
local M4 l.wBeqta
local M5 l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4 

* Logit coeff. 
logit YBCId_release_ext `M1' Dcountries* if D_FILRtime_n2==1,   cluster(banks)
est sto m
outreg2 using H2.Results.csv, replace bdec(3) sdec(3)  label addtext("Year fixed effects", NO, "Country fixed effects", YES) addstat("Pseudo R-squared", e(r2_p), "Log-likelihood", e(ll)) keep(`M4' `M5') groupvar(Catch_ratios `M4' Controls `M5') cti(Logit coeff.)

logit YBDGd_release_ext `M1' Dcountries* if D_FILRtime_n2==1,   cluster(banks)
est sto m
outreg2

* test of coeff Logit
logit YBCId_release_ext `M1' Dcountries* if D_FILRtime_n2==1
est sto m1
logit YBDGd_release_ext `M1' Dcountries* if D_FILRtime_n2==1
est sto m2
suest m1 m2
test [m1_YBCId_release_ext]L.wBeqta  = [m2_YBDGd_release_ext]L.wBeqta


* MLM Coefficients
mlogit Trig_release_ext `M1' Dcountries* if D_FILRtime_n2==1,   cluster(banks) iterate(10)  
est sto m
outreg2 using H2.Results.csv, append bdec(3) sdec(3)  label addtext("Year fixed effects", NO, "Country fixed effects", YES) addstat("Pseudo R-squared", e(r2_p), "Log-likelihood", e(ll)) keep(`M4' `M5') groupvar(Catch_ratios `M4' Controls `M5') cti(MLN coeff.)
mlogit
* test of coeff MLN
mlogit Trig_release_ext `M1' Dcountries* if D_FILRtime_n2==1,   cluster(banks) iterate(10)  
test [Liquidity_release]L.wBeqta  = [Capital_release]L.wBeqta


* Ordered Logit odds ratio
ologit Trig_release_ext_o2 `M1' Dcountries* if D_FILRtime_n2==1,   cluster(banks) 
est sto m
outreg2 using H2.Results.csv, append bdec(3) sdec(3)  label addtext("Year fixed effects", NO, "Country fixed effects", YES) addstat("Pseudo R-squared", e(r2_p), "Log-likelihood", e(ll)) keep(`M4' `M5') groupvar(Catch_ratios `M4' Controls `M5') cti(Ordered Logit coeff.)





****************************************************************************************************************************************************************
* Table 8
****************************************************************************************************************************************************************

* H3: Harshness of restrictions and Equity to Total assets

local M1 l.wBlcd l.wln_Btas l.wBcredit_risk l.wBroaa l.wBroaa_std4 

* OLS regressions: Release trigger Capital injections
xtreg wBeqta ipca_cio5 `M1'     if  f.BCI_release_ext==1,   cluster(banks)
outreg2 using H3.Results_restrictions_CI.csv, replace bdec(3) sdec(3)  label addtext("Year fixed effects", NO, "Country fixed effects", NO) addstat("Within R-squared", e(r2_w)) keep(ipca_cio5 iew_cio5 `M1') groupvar(Restrictiveness_measures ipca_cio5 iew_cio5 Controls `M1')

xtreg wBeqta iew_cio5 `M1'     if  f.BCI_release_ext==1,   cluster(banks)
outreg2, append


* OLS regressions: Release trigger Debt guarantees 
xtreg wBeqta ipca_dgo5 `M1'     if  f.BDG_release_ext==1,   cluster(banks)
outreg2 using H3.Results_restrictions_DG.csv, replace bdec(3) sdec(3)  label addtext("Year fixed effects", NO, "Country fixed effects", NO) addstat("Within R-squared", e(r2_w)) keep(ipca_dgo5 iew_dgo5 `M1') groupvar(Restrictiveness_measures ipca_dgo5 iew_dgo5 Controls `M1')

xtreg wBeqta iew_dgo5 `M1'     if  f.BDG_release_ext==1,   cluster(banks)
outreg2, append





****************************************************************************************************************************************************************
* Table 9
****************************************************************************************************************************************************************

* H4: Harshness of restrictions and duration of bailouts

* Capital bailouts
local M1 ipca_cio5 iew_cio5 
local M4 RCspread_Beqta_t_1 l_wBlcd l_wln_Btas l_wBcredit_risk l_wBroaa l_wBroaa_std4

* Cox model censored
* extended sample
stset duration_CI, failure(status_BCI)

stcox ipca_cio5 `M4' if  BCI_release_ext==1,   vce(robust) nohr 
outreg2 using H4.Results_duration_CI.csv, replace bdec(3) sdec(3)  label addtext("Year fixed effects", NO, "Country fixed effects", NO) addstat("Pseudo R-squared", e(r2_p), " Wald chi2", e(chi2), "Log-likelihood", e(ll)) keep(`M1' `M4') groupvar(Restrictions `M1' Controls `M4') cti(Cox model censored: extended sample)
stcox iew_cio5 `M4' if  BCI_release_ext==1,   vce(robust) nohr 
outreg2


* Debt guarantee bailouts
local M1 ipca_dgo5 iew_dgo5 
local M4 RCspread_Beqta_t_1 l_wBlcd l_wln_Btas l_wBcredit_risk l_wBroaa l_wBroaa_std4

* Cox ,odel censored
* extended sample
stset duration_DG, failure(status_BDG)

stcox ipca_dgo5 `M4' if  BDG_release_ext==1,   vce(robust) nohr 
outreg2 using H4.Results_duration_DG.csv, replace bdec(3) sdec(3)  label addtext("Year fixed effects", NO, "Country fixed effects", NO) addstat("Pseudo R-squared", e(r2_p), " Wald chi2", e(chi2), "Log-likelihood", e(ll)) keep(`M1' `M4') groupvar(Restrictions `M1' Controls `M4') cti(Cox model censored: extended sample)
stcox iew_dgo5 `M4' if  BDG_release_ext==1,   vce(robust) nohr 
outreg2





****************************************************************************************************************************************************************
* Table 10
****************************************************************************************************************************************************************

* H5-H7: Bank characteristics and the choice of bailout method

* LPM
local M4 l.wln_Btas l.wPbtas_ctas l.wBdepta l.wBdepliab l.wBroaa_std4 l.wBroae_std4 
local M5 l.wBlcd l.wBcredit_risk l.wBroaa


local M1 l.wln_Btas l.wBdepta l.wBroaa_std4 l.wBlcd l.wBcredit_risk l.wBroaa
xtreg Bailout_2trig_all_o_d `M1'  Dyear* if year>=2008, fe  vce(cluster banks)
outreg2 using H5-H7.Results.csv, replace stats(coef tstat) bdec(3) tdec(3)  label addtext("Bank fixed effects", YES) addstat("Adjusted R-squared", e(r2_a)) keep(`M4' `M5') groupvar(Bank_characteristics `M4' Controls `M5') cti(LPM)

local M1 l.wln_Btas l.wBdepta l.wBroaa_std4 l.wBlcd l.wBcredit_risk l.wBroaa
xtreg Bailout_2trig_all_o_d `M1'  Dyear* if year>=2008 & Intervened_2triggers_both==1, fe  vce(cluster banks)
outreg2 using H5-H7.Results.csv, append stats(coef tstat) bdec(3) tdec(3)  label addtext("Bank fixed effects", YES) addstat("Adjusted R-squared", e(r2_a)) keep(`M4' `M5') groupvar(Bank_characteristics `M4' Controls `M5') cti(LPM)


local M1 l.wPbtas_ctas l.wBdepliab l.wBroae_std4 l.wBlcd l.wBcredit_risk l.wBroaa
xtreg Bailout_2trig_all_o_d `M1'  Dyear* if year>=2008, fe  vce(cluster banks)
outreg2 using H5-H7.Results.csv, append stats(coef tstat) bdec(3) tdec(3)  label addtext("Bank fixed effects", YES) addstat("Adjusted R-squared", e(r2_a)) keep(`M4' `M5') groupvar(Bank_characteristics `M4' Controls `M5') cti(LPM)

local M1 l.wPbtas_ctas l.wBdepliab l.wBroae_std4 l.wBlcd l.wBcredit_risk l.wBroaa
xtreg Bailout_2trig_all_o_d `M1'  Dyear* if year>=2008 & Intervened_2triggers_both==1, fe  vce(cluster banks)
outreg2 using H5-H7.Results.csv, append stats(coef tstat) bdec(3) tdec(3)  label addtext("Bank fixed effects", YES) addstat("Adjusted R-squared", e(r2_a)) keep(`M4' `M5') groupvar(Bank_characteristics `M4' Controls `M5') cti(LPM)






